clear all
set more off
set matsize 2000
set seed 20240214

********************************************************************
* Create a fake dataset that replicates the format of MSCE data

set obs 1508
gen ID = _n

* Four schools 
gen school = 1 if ID <=400 
replace school = 2 if ID>400 & ID<=800
replace school = 3 if ID>800 & ID<=1154
replace school = 4 if ID>1154 & ID<=1508

* PID 
bys school: gen pid = _n
replace pid = pid + 400

* FID (schoolID + PID)
gen fid = .
forval y = 1/4 {
	replace fid = pid + `y'*1000 if school == `y'
}

* Form 
gen form = .
replace form = 2 if pid >=400 & pid <=525
replace form = 3 if pid >525 & pid <=650
replace form = 4 if form == . // this does not necessarily replicates the sample distribution in our data. 

* Treatement
bys school form: gen ra_T = rbinomial(1, 0.2) // 1/5 of students are in the T group

* Grade - below/above median
bys school form ra_T: gen lgrade = rbinomial(1, 0.5) 
gen T_ra_strat_lgrade = ra_T*lgrade
gen T_ra_strat_hgrade = ra_T*(1 -lgrade)

* Baseline English grade
gen gr0_eng_missing = rbinomial(1, 0.04) // in our data we have 4.2% missing baseline English grades
gen gr0_eng_0 = rnormal(0,1)

* gender
gen male = 0 if school == 1 
replace male = 1 if school == 2 
replace male = runiformint(1,2) if male==.

* Strata
bys school form male lgrade: gen strata = _n 
replace strata = strata - 50 if strata>=52
tab strata // (we have 51 strata in the original intervention)

* MSCE scores
gen temp_att = rbinomial(1, 0.05)

foreach out in addma agr bio bk chem chi com drw eng fre geo he his ls mat mw phy ss {
	gen msce_`out' = runiformint(1,9)
	replace msce_`out' = . if temp_att==1 // for attrition - we suppose 5% of students do not complete MSCE, switched schools, or took their exam at a later year. 
}

* Less students normally complete the following msce exams so replace with missings as well. 

foreach out in agr geo ls ss {
	gen temp_miss`out' = rbinomial(1, 0.3) // these prob. numbers are based on data from Form 4 students which was already in possession and previously analysed.
	replace msce_`out' = . if temp_miss`out'==1 
	drop temp_miss`out'
}

foreach out in bk com he {
	gen temp_miss`out' = rbinomial(1, 0.8)
	replace msce_`out' = . if temp_miss`out'==1 
	drop temp_miss`out'
}

foreach out in addma chem drw fre his mw {
	gen temp_miss`out' = rbinomial(1, 0.95)
	replace msce_`out' = . if temp_miss`out'==1 
	drop temp_miss`out'
}

******** THIS ENDS THE CODE TO CREATE A FAKE DATASET THAT REPLICATES THE INFORMATION WE WILL OBTAIN. 
********************************************************************


********************Q: Does the intervention affect MSCE pass rates and scores?****************************************************

*For Form 4 we use administrative data (already in possession and previously analysed)
*For Forms 2 and 3 we plan to use administrative data if available, otherwise we will use survey data for students with missing administrative data.
*We anticipate that administrative data will be missing by entire classes and/or schools, so there is no risk of differential attrition in this particular administrative data

*Here we are always splitting by high/low achievers, as we do not expect any effect on high achievers

local covars "gr0_eng_0 gr0_eng_missing"

*Calculate pass rate

*Obtaining a pass in six subjects, including English, with at least one distinction/ credit

*Or obtaining a pass in five subjects, including English, with at least three distinctions/ credits
foreach x of varlist msce_* {
gen `x'_pass=0
gen `x'_credit=0
replace `x'_pass=1 if `x'<9 
replace `x'_credit=1 if `x'<=6 
}

egen num_pass=rowtotal(*_pass)
egen num_credit=rowtotal(*_credit)

gen pass=0

*If form 4 student did not write exams then they did not pass. Some students might have switched schools or take MSCE another and later year. We will address this concern with survey data as a robustness check. 

*replace pass=. where all MSCE data is missing for the whole school etc
replace pass=1 if num_pass>=6 & num_pass!=.
replace pass=0 if msce_eng>=9
replace pass=0 if num_credit==0

*Or obtaining a pass in five subjects, including English, with at least three distinctions/ credits
replace pass=1 if msce_eng<9 & num_pass>=5 & num_credit>=3

*Score required for admission to public university
*replace admission=. where all MSCE data is missing for the whole school etc
gen admission=0
replace admission=1 if num_credit>=6
replace admission=0 if msce_eng>=9

*Calculate MSCE aggregate
*1 = distinction, 2 = distinction, 3 = credit, 4 = credit, 5 = credit, 6 = credit, 7 = pass, 8 = pass, 9 = fail

gen used=0
gen temp=0
gen agg=0
egen numMSCEexams = rownonmiss(msce_addma-msce_ss)

foreach x of varlist msce* {
		gen run_`x' = `x'
				
} 

forvalues i=1/6 {
	
gen top`i'=.
foreach x of varlist run_msce* {
		replace top`i'=`x' if top`i'>`x' // Creates the lowest grades (best score)
}

replace used=0
replace temp=0
foreach x of varlist msce* {
		replace temp=1 if top`i'==run_`x' & used==0
		replace run_`x'=. if top`i'==run_`x' & used==0
		replace used=1 if temp==1
}
replace agg=agg+top`i' // Total of best 6 scores. 
}

replace agg=. if numMSCEexams<6 // Needs 6 grades to pass.
replace agg=. if msce_eng>8 // If failed english, do not pass.
rename agg msceagg
drop run* used temp

* Because a lower score is associated with a better performance, we transform the msce_score for an easier interpretation. 
foreach x of varlist msce_eng msce_bio msce_mat msce_chi {
	replace `x' = 10 - `x'
}

*Run regressions, always splitting by early/late and low/high achievers

foreach x of varlist admission pass msceagg msce_eng msce_bio msce_mat msce_chi {
	
	*For same-year intervention (fake data)
	reghdfe `x' T_ra_strat_lgrade T_ra_strat_hgrade `covars' if form==4, absorb(strata) vce(robust) noconstant
	
	*For those who got an earlier intervention (forms 2 and 3, fake data)
	reghdfe `x' T_ra_strat_lgrade T_ra_strat_hgrade `covars' if form<4, absorb(strata) vce(robust) noconstant

	}




